```
$EOLCOM >

$OffsYMLIST OffLISTING

OPTION LIMROW= 0,LIMCOL = 0;

$ontext

A FIMM Program of Efficient Crop Selection & Water Use in Texas Lower Colorado River Basin

Programmed by Frank A. Ward
New Mexico State University
Las Cruces, NM  88003  USA
(505) 646-1220
em: fward@nmsu.edu

most recent programming June 18, 2006

Minimizes taxpayer cost of farm subsidies to irrigators needed to maintain farm income at 'without project' level
while reducing water supply for agriculture 'with project'

With project allows for high yielding rice varieties and subsidies of water conservation land improvement measures

Without project allows for neither

TABLE OF CONTENTS

   Section 1. Set specification
   Section 2. Resource use coefficients
   Section 3. yield, price, cost, and net return data
   Section 4. Right-hand-side resource limits
   Section 5. Variable and equation specification
   Section 6. Displays (post optimality)

$offtext

**************** Section 1 ************************
*  The following sets are specified as indices     *
*  for variables and parameters.                   *
*  Correspondence sets are also specified to       *
*  limit allowable set element combinations        *
*  Some combinations may not exist.                *
****************************************************


SETS

c  conservation measure    / base, m_in, p_gr              /
j  crops                   / cv_rice, hy_rice              /
k  district                / Lakeside-M                    /
l  crop season             / sp_su, su_fa                  /
m  water source            / sw                            /
n  inputs                  / water, other                  /
t  time period             / 2005 * 2010                   /
p  project: with v without / wop  without,  wip  with proj /

jp(j,p)  allowed crop to project combinations: hy rice not allowed without project

/
    cv_rice  .wop

    cv_rice  .wip
    hy_rice  .wip
/
```

```
cc(c)    any cons measure other than base
cb(c)    base conservation measure
tf(t)    1st period only
tt(t)    any period after 1st pd
dc(t)    capital renewal every decade (10th year) beginning with year 1
nd(t)    all periods except decade years
;

cc(c)  = yes $ (ord(c) gt 1);         > gams language picks every measure after base
cb(c)  = yes $ (ord(c) eq 1);         > gams lanaguge picks only base measure

tf(t)  = yes $ (ord(t) eq 1);         > gams language picks first period only
tt(t)  = yes $ (ord(t) gt 1);         > gams language picks every period after 1st
*dc(t) $ (mod(ord(t),10) eq 1) = yes;  > gams language picks every 10th year beginning with year 1 - 11 - etc
*nd(t)  = not dc(t);                   > gams language picks every year except years 1 - 11 - etc

*display cc;
*display cb;
*display tf;
*display tt;
*display dc;  > GAMS code thanks to Juha Haataja Scientific Computing Ltd Finland: jhaataja@csc.fi
*display nd;

$ontext

c   conservation measures  base (no conservation), m_inlets (multiple inlets)
j   crops:                 cv_rice, hy_rice
k   district               Lakeside-M,
l   crop season            sp_su 1st crop;  su_fa  2nd crop;
m   water source:          sw   = surface water,  gw = groundwater
n   inputs                 inputs used in NMSU extension budgets
t   time period            1 * n years
p   project                wop  = without project, wip  with project


project is defined this way:

1.  providing high yield rice varieties
2.  providing access to water conservation measures at LSWP subsidized prices.

No compensation will occur for producers who reduce baseline water use through:

1.  reduced acreage
2.  better water management
3.  no second crop
4.  exiting the industry
5.  any other decisions than adopting high yield rice varieties or adopting subsidized water conserving technologies

project is NOT credited with water savings

1.  when producers conserve water with unsubsidized base technology
2.  when producers conserve water with unsubsidized water conserving technology
3.  when producers switch to water conserving technology without the project
4.  when producers avoid switching to water conserving technology with the project

Without project means

1.  conventional rice varieties
2.  baseline water conservation measures and their costs, including all non LSWP subsidies (e.g., TX or Fed)

With project means

1.  higher yield rice varieties
```

```
2.  lower prices for water conserving measures by LSWP than with all other subsidies

without project means: Producers maximize net income while facing the following conditions:

    a.  conventional rice varieties
    b.  baseline unsubsidized conservation practice (simple conservation tillage)
    c.  no LSWP subsidies for any alternative conservation practice
    d.  surface water supply availabile for ag from Brandes data

    e.  include all conservation practice subsidies from TX State (HB 1443) and US federal (EQUIP) cost share programs
    f.  current LCRA surface water prices and shortage allocation rules
    g.  best estimate of current and future depth to groundwater
    h.  best available rice price forecast (loan rate + premium + outside income (37/acre)
    i.  best available forecast of other input prices

with project means:  Producers maximize net income while facing the following conditions:

    a.  conventional or high yield rice varieties (producer chooses)
    b.  baseline unsubsidized conservation practices (simple conservation tillage)
    c.  subsidies for all alternative conservation practices
    d.  surface water supply availabile for ag from Brandes data minus 40,000 acre feet per year summed over 3 districts

    e.  include all conservation practice subsidies from TX State (HB 1443) and US federal (EQUIP) cost share programs
    f.  current LCRA surface water prices and shortage allocation rules
    g.  best estimate of current and future depth to groundwater
    h.  best available rice price forecast (loan rate + premium + outside income (37/acre)
    i.  best available forecast of other input prices

$offtext

**************  SECTION 2 *********************
***  Resource Use coefficients ****************
***********************************************

Table Wateruse_p(c,j,k,l,m,p)
                                      wop    wip
base.cv_rice.Lakeside-M.sp_su.sw     2.80   2.80
base.cv_rice.Lakeside-M.su_fa.sw     1.00   1.00
base.hy_rice.Lakeside-M.sp_su.sw     2.86   2.86
base.hy_rice.Lakeside-M.su_fa.sw     1.02   1.02

m_in.cv_rice.Lakeside-M.sp_su.sw     2.10   2.10
m_in.cv_rice.Lakeside-M.su_fa.sw     0.75   0.75
m_in.hy_rice.Lakeside-M.sp_su.sw     2.14   2.14
m_in.hy_rice.Lakeside-M.su_fa.sw     0.77   0.77

p_gr.cv_rice.Lakeside-M.sp_su.sw     2.07   2.07
p_gr.cv_rice.Lakeside-M.su_fa.sw     0.74   0.74
p_gr.hy_rice.Lakeside-M.sp_su.sw     2.12   2.12
p_gr.hy_rice.Lakeside-M.su_fa.sw     0.76   0.76
;

* water use is acre feet per acre per season
* various web sources for rice in TX, LA, MS, and CA

options wateruse_p: 2:0:1;
display wateruse_p;

*$ontext

Parameter Landuse_p(c,j,k,l,m,t,p)           ;
          Landuse_p(c,j,k,l,m,t,p)        = 1;

*land use is acres per acre per season
```

```
************  section 3  *****************************************
*  yields, prices, costs, and net returns per acre  ***************
******************************************************************

Table Yield_p(c,j,k,l,p) Expected yields: L. Ted Wilson TAMU AES Beaumont Center (June 2006) hy rice only with project

                                         wop         wip
base.cv_rice.Lakeside-M.sp_su             70          70
base.cv_rice.Lakeside-M.su_fa             21          21
base.hy_rice.Lakeside-M.sp_su                         91
base.hy_rice.Lakeside-M.su_fa                         27

m_in.cv_rice.Lakeside-M.sp_su             70          70
m_in.cv_rice.Lakeside-M.su_fa             21          21
m_in.hy_rice.Lakeside-M.sp_su                         91
m_in.hy_rice.Lakeside-M.su_fa                         27

p_gr.cv_rice.Lakeside-M.sp_su             70          70
p_gr.cv_rice.Lakeside-M.su_fa             21          21
p_gr.hy_rice.Lakeside-M.sp_su                         91
p_gr.hy_rice.Lakeside-M.su_fa                         27
;
*Jensen budgets: 69.3 cwt 1st crop; 16.8 cwt 2nd crop

options Yield_p: 2:0:1;
display Yield_p;

scalar epsi
/0.001/

Table Pric_p(j,l)  Rice Price (loan) from 2004 LCRA (Jensen) Budgets - based on TAMU CES

                  sp_su     su_fa
cv_rice           7.75      7.75
hy_rice           7.75      7.75
;

* rice price is loan rate dollars per cwt;

Table Premium_p(j,l)   Rice Price Premium from 2004 LCRA Budgets - based on TAMU CES

                  sp_su     su_fa
cv_rice           2.00      2.00
hy_rice           2.00      2.00
;

Table Direct_p(j,l)     Direct government rice payment per cwt when cwt > 55.

                  sp_su     su_fa
cv_rice           2.35         0
hy_rice           2.35         0
;

Parameter Pric_p(j,l)  Total rice price per cwt = loan + premium +  Direct payments
;
Pric_p(j,l) = Pric_p(j,l) + Premium_p(j,l) + Direct_p(j,l);

*display pric_p;

scalar pg   annual rice price growth rate -- positive .01 means + 1% growth rate per year

/-.02  /
;
```

```
Parameter Price_p(j,l,t)   crop price by crop time and season including loan + premium + direct payments
;
Price_p(j,l,t) = Pric_p(j,l) * (1 + pg) ** (ord(t));

options Price_p:2:0:1;
display Price_p;

table   oth_inc_p(j,l)    Outside income per acre: source LCRA 2004 Budgets divides 50 50 into 1st and 2nd crops - same for cv and hy

                  sp_su     su_fa
cv_rice            18.67    18.67
hy_rice            18.67    18.67
;

Table W_Price_p(c,j,k,l,m,   p)   Water Price Per Acre Foot: Source Kyle Jensen LCRA 2004 budgets (2 pt water price - pr acre and pr af)

                                           wop         wip
base.cv_rice.Lakeside-M.sp_su.sw         10.96       10.96
base.cv_rice.Lakeside-M.su_fa.sw          9.86        9.86
base.hy_rice.Lakeside-M.sp_su.sw         10.96       10.96
base.hy_rice.Lakeside-M.su_fa.sw          9.86        9.86

m_in.cv_rice.Lakeside-M.sp_su.sw         10.96       10.96
m_in.cv_rice.Lakeside-M.su_fa.sw          9.86        9.86
m_in.hy_rice.Lakeside-M.sp_su.sw         10.96       10.96
m_in.hy_rice.Lakeside-M.su_fa.sw          9.86        9.86

p_gr.cv_rice.Lakeside-M.sp_su.sw         10.96       10.96
p_gr.cv_rice.Lakeside-M.su_fa.sw          9.86        9.86
p_gr.hy_rice.Lakeside-M.sp_su.sw         10.96       10.96
p_gr.hy_rice.Lakeside-M.su_fa.sw          9.86        9.86
;

options W_price_p:2:0:1;
display W_price_p;

$ontext

WATER PRICE SOURCE:  Kyle Jensen's LCRA 2004 cost and return budgets
water has a 2 part price:  (1) price per acre (1st crop only),
                           (2) price per af   (1st and 2nd crop)

ITEM                                AMOUNT        UNITS
1st Crop Water Application          2.8         ac-ft/acre
2nd Crop Water Application          1.0         ac-ft/acre
Base Water Charge - 1st Crop      $39.20         per acre
Volume Water Charge - 1st crop    $10.96         per ac-ft
Volume Water Charge - 2nd crop     $9.86         per ac-ft

$offtext


Table Inputcost_a(c,j,k,l,m,n,p) Input Prodxn Costs PER ACRE:  Kyle Jensen LCRA base budgets incl water charge per acre but not per af

                                   water.wop      other.wop         water.wip       other.wip
base.cv_rice.Lakeside-M.sp_su.sw     39.20         712.60            39.20          712.60
base.cv_rice.Lakeside-M.su_fa.sw         0         116.51                0          116.51
base.hy_rice.Lakeside-M.sp_su.sw     39.20         712.60            39.20          712.60
base.hy_rice.Lakeside-M.su_fa.sw         0         116.51                0          116.51

m_in.cv_rice.Lakeside-M.sp_su.sw     39.20         712.60            39.20          712.60
m_in.cv_rice.Lakeside-M.su_fa.sw         0         116.51                0          116.51
m_in.hy_rice.Lakeside-M.sp_su.sw     39.20         712.60            39.20          712.60
```

```
m_in.hy_rice.Lakeside-M.su_fa.sw         0         116.51                 0          116.51

p_gr.cv_rice.Lakeside-M.sp_su.sw     39.20         712.60             39.20          712.60
p_gr.cv_rice.Lakeside-M.su_fa.sw         0         116.51                 0          116.51
p_gr.hy_rice.Lakeside-M.sp_su.sw     39.20         712.60             39.20          712.60
p_gr.hy_rice.Lakeside-M.su_fa.sw         0         116.51                 0          116.51
;

scalar rho interest-discount rate
/0.05/

parameter rat_pct_p(k)   maximum percentage of acres in ratoon crop (historical data yr 2005)

/Lakeside-M     0.76/


Parameter Inputcost_p(c,j,k,l,m,n,t,p);
Inputcost_p(c,j,k,l,m,n,t,p) = inputcost_a(c,j,k,l,m,n,p);

options Inputcost_p: 2:0:1;
*Display Inputcost_p;

parameter DF(t)  discount rate factor;

DF(t) = 1/((1+rho) ** (ord(t)-1));

*display df;

Table C_Cost_p(c,j,k,l,      p)      up front added capital cost of cons measures per acre compared to base

                                  wop          wip
base.cv_rice.Lakeside-M.sp_su       0            0
base.cv_rice.Lakeside-M.su_fa       0            0
base.hy_rice.Lakeside-M.sp_su       0            0
base.hy_rice.Lakeside-M.su_fa       0            0

m_in.cv_rice.Lakeside-M.sp_su      80           80
m_in.cv_rice.Lakeside-M.su_fa      80           80
m_in.hy_rice.Lakeside-M.sp_su      80           80
m_in.hy_rice.Lakeside-M.su_fa      80           80

p_gr.cv_rice.Lakeside-M.sp_su     250          250
p_gr.cv_rice.Lakeside-M.su_fa     250          250
p_gr.hy_rice.Lakeside-M.sp_su     250          250
p_gr.hy_rice.Lakeside-M.su_fa     250          250

* data source consulting report dated June 6, 2006, by Ken Kuhr on costs of wat con measures
;

table life_p(c,j,k,l,    p)          years life of land improvement

                                  wop          wip
base.cv_rice.Lakeside-M.sp_su       1            1
base.cv_rice.Lakeside-M.su_fa       1            1
base.hy_rice.Lakeside-M.sp_su       1            1
base.hy_rice.Lakeside-M.su_fa       1            1

m_in.cv_rice.Lakeside-M.sp_su      15           15
m_in.cv_rice.Lakeside-M.su_fa      15           15
m_in.hy_rice.Lakeside-M.sp_su      15           15
m_in.hy_rice.Lakeside-M.su_fa      15           15

p_gr.cv_rice.Lakeside-M.sp_su      20           20
p_gr.cv_rice.Lakeside-M.su_fa      20           20
```

```
p_gr.hy_rice.Lakeside-M.sp_su      20           20
p_gr.hy_rice.Lakeside-M.su_fa      20           20
;

parameter am_factor (c,j,k,l,   p)   amortization factor: amortizes a one shot capital investment as a fn of life and interest rate
parameter A_C_Cost_p(c,j,k,l, t,p)   amortized annual capital cost per year of land improvement by c-t-p
;
am_factor (c,j,k,l,           p) =    rho / (1 - ((1 + rho) ** (-life_p(c,j,k,l,p))));
A_C_cost_p(c,j,k,l,         t,p) =        C_cost_p(c,j,k,l,p) * am_factor(c,j,k,l,p   );

*display am_factor, A_C_cost_p;

*$ontext

parameter  C_C_p(c,j,k,l,m,t,p)      amortized annual capital cost per year of land improvement for all sets
;

C_c_p( c,j,k,l,m,t, p) = A_C_Cost_p(c,j,k,l,t,p);
C_c_p(cb,j,k,l,m,tt,p) = 0;

*options C_c_p:2:0:1;
*display C_c_p;

Table  O_Cost_p(c,j,k,l,      p)       incremental operation cost of cons measure per acre per year compared to base measure

                                         wop       wip
base.cv_rice.Lakeside-M.sp_su              0         0
base.cv_rice.Lakeside-M.su_fa              0         0
base.hy_rice.Lakeside-M.sp_su              0         0
base.hy_rice.Lakeside-M.su_fa              0         0

m_in.cv_rice.Lakeside-M.sp_su             12        12
m_in.cv_rice.Lakeside-M.su_fa             12        12
m_in.hy_rice.Lakeside-M.sp_su             12        12
m_in.hy_rice.Lakeside-M.su_fa             12        12

p_gr.cv_rice.Lakeside-M.sp_su             50        50
p_gr.cv_rice.Lakeside-M.su_fa             50        50
p_gr.hy_rice.Lakeside-M.sp_su             50        50
p_gr.hy_rice.Lakeside-M.su_fa             50        50
;

Parameter O_c_p(c,j,k,l,m,t,p);
O_c_p(c,j,k,l,m,t, p) = O_Cost_p(c,j,k,l,   p);

*options O_c_p:2:0:1;
*display O_c_p;


Table  L_Cost_p(c,j,k,l,    p)       incremental labor and other cost of cons meas per acre per year compared to base measure

                                    wop       wip
base.cv_rice.Lakeside-M.sp_su         0         0
base.cv_rice.Lakeside-M.su_fa         0         0
base.hy_rice.Lakeside-M.sp_su         0         0
base.hy_rice.Lakeside-M.su_fa         0         0

m_in.cv_rice.Lakeside-M.sp_su         0         0
m_in.cv_rice.Lakeside-M.su_fa         0         0
m_in.hy_rice.Lakeside-M.sp_su         0         0
m_in.hy_rice.Lakeside-M.su_fa         0         0

p_gr.cv_rice.Lakeside-M.sp_su    -16.50    -16.50
p_gr.cv_rice.Lakeside-M.su_fa    -16.50    -16.50
```

```
p_gr.hy_rice.Lakeside-M.sp_su    -16.50    -16.50
p_gr.hy_rice.Lakeside-M.su_fa    -16.50    -16.50
;

Parameter L_c_p(c,j,k,l,m,t,p);
L_c_p(c,j,k,l,m,t, p) = L_Cost_p(c,j,k,l,   p);

*options L_c_p:2:0:1;
*display L_c_p;

Parameter Costacre_p(c,j,k,l,m,t,p)  total input cost per acre EXCLUDING water conservation subsidies
          Watcost_p (c,j,k,l,m,  p)  total water cost per acre based on price pr af * af pr ac
          TotRev_p  (c,j,k,l,  t,p)  total revenue per acre = sum of all prices * yield
          TotCos_p  (c,j,k,l,m,t,p)  total cost per acre including capital and operating costs for cons measures
          Netrev_p  (c,j,k,l,m,t,p)  total net revenue per acre incl all costs of cons measures = tot rev - total cost
;

Costacre_p(c,j,k,l,m,t,p)  = sum(n,   Inputcost_p(c,j,k,l,m,n,t,p));

Watcost_p (c,j,k,l,m,  p)  = W_Price_p(c,j,k,l,m,p) * wateruse_p(c,j,k,l,m,p);

TotRev_p  (c,j,k,l,  t,p)  =   Price_p(j,l,t) * Yield_p(c,j,k,l,      p)
                           + Oth_inc_p(j,l);

TotCos_p  (c,j,k,l,m,t,p)  =           Costacre_p(c,j,k,l,m,  t,p)
                                     +  Watcost_p(c,j,k,l,m,    p)
                                     +      C_c_p(c,j,k,l,m,  t,p)
                                     +      O_c_p(c,j,k,l,m,  t,p)
                                     +      L_c_p(c,j,k,l,m,  t,p)
;
Netrev_p  (c,j,k,l,m,t,p)  =             TotRev_p  (c,j,k,l,  t,p)
                                      -  TotCos_p  (c,j,k,l,m,t,p)
;

options TotRev_p: 2:0:1;
options Totcos_p: 2:0:1;
options netrev_p: 2:0:1;
display TotRev_p, TotCos_p, Netrev_p;

***************************************
******   section 4            *********
*****    rhs resource limits  *********
***************************************

Parameter
Landrhs_a(k)         Total irrigable acres by district
S_waterrhs_a(k)      total surface water supplies by district
Wat_sav(k)           Required water savings by district in acre feet per year with project

Landrhs_a(k)
/
Lakeside-M             26360
/

S_waterrhs_a(k)
/
Lakeside-M             93691
/

Wat_sav(k)
/
Lakeside-M             15000
/
;
```

```
Parameter
Landrhs_p   (k,l,t,p)       land supply limit
S_Waterrhs_p(k,  t,p)       water supply limit
;

Landrhs_p   (k,l,t,p)    = Landrhs_a(k);
S_Waterrhs_p(k,  t,p)    = S_Waterrhs_a(k);


scalar wg   annual water supply growth rate -- positive .01 means + 1% growth rate per year

/-.01  /
;

Parameter S_Waterrhs_p(k,   t,p)    Surface water supply by district time and project
;
S_Waterrhs_p(k,   t,p) = S_Waterrhs_a(k) * (1 + pg) ** (ord(t));

* reduce surface water with the project by X acre feet
S_Waterrhs_p(k,t,'wip') = S_Waterrhs_p(k,t,'wop') - Wat_sav(k);

options landrhs_p:2:0:1;
options S_waterrhs_p:2:0:1;
display Landrhs_p, S_waterrhs_p;

*****************************************
********  section 5  *******************
********** variables and equations ******
*****************************************

Variables
  Acres_v (c,j,k,l,m, t,p)   Acres of jth crop   (by crop-capital-water level-water source)

Netrv_2_v (c,j,k,l,m, t,p)   Net revenue per acre incl con meas subsidy for acres in prodn          (PRECISE)
Subac_2_v (c,j,k,l,m, t,p)   SUBSIDY PER ACRE of capital + op cons meas for acres in prodn          (PRECISE)
 C_c_2_v  (c,j,k,l,m, t,p)   total cap cost cons meas per acre including subsidy for acres in prodn (PRECISE)
 C_c_s_2_v(c,j,k,l,m, t,p)   SUBSIDY per acre: capital cost cons measure                            (PRECISE)
 O_c_2_v  (c,j,k,l,m, t,p)   total op cost cons meas per acre including subsidy for acres in prodn  (PRECISE)
 O_c_s_2_v(c,j,k,l,m, t,p)   SUBSIDY per acre: operating cost cons measure                          (PRECISE)

 Netrev_v (c,j,k,l,m, t,p)   Net revenue per acre including cons meas subsidy                          (IMPRECISE)
Subacre_v (c,j,k,l,m, t,p)   SUBSIDY PER ACRE incl capital + operation cost cons measure               (IMPRECISE)
 C_c_v    (c,j,k,l,m, t,p)   capital cost cons meas per acre  INCLUDING SUBSIDY for acres in prodn     (IMPRECISE)
 C_c_s_v  (c,j,k,l,m, t,p)   SUBSIDY per acre: capital cost cons measure                               (IMPRECISE)
 O_c_v    (c,j,k,l,m, t,p)   operating cost cons measure per acre INCLUDING SUBSIDY for acres in prodn (IMPRECISE)
 O_c_s_v  (c,j,k,l,m, t,p)   SUBSIDY per acre: operating cost cons measure                             (IMPRECISE)


Incom_a_v (c,j,k,l,m, t,p)   Farm Income over acres for each index
Incom_c_v (  j,k,l,m, t,p)   Farm Income over cons measures for each index
Income_v  (           t,p)   Farm Income by period - with and without project

NPV_v     (             p)   NPV farm income WITHOUT and WITH project
NPV_wop_v                    NPV farm income WITHOUT project
NPV_wip_v                    NPV farm income WITH    project

subsd_a_v (c,j,k,l,m, t,p)   Total taxpayer cost of subsidy over acres for each index
P_cost_v  (             p)   NPV taxpayer cost conservation program  WITH and WITHOUT project
P_cos_wip_v                  NPV taxpayer cost conservation programs WITH project

;
Positive Variables
  Acres_v  (c,j,k,l,m,t,p)
```

```
  Subacre_v (c,j,k,l,m, t,p)
  subsd_a_v (c,j,k,l,m, t,p)
  P_cost_v  (             p)

;

Equations

S_Water_e (    k,     t,p)   Obey surface water constraint by time and project
Land_e    (    k,l,   t,p)   Obey land constraint by time and project
up_front  (c,j,k,l,m, t,p)   Obey up front capital costs reqd for conservation improvement
Season_e  (c,j,k,l,m, t,p)   Acres of ratoon crop can never exceed acres of base crop

Netrv_2_e (c,j,k,l,m, t,p)   Net revenue per acre with INCLUDING subsidy (PRECISE)
Subac_2_e (c,j,k,l,m, t,p)   capital + operating cost subsidy per acre   (PRECISE)
C_c_2_e   (c,j,k,l,m, t,p)   capital cost per acre INCLUDING subsidy     (PRECISE)
C_c_s_2_e (c,j,k,l,m, t,p)   capital cost SUBSIDY per acre               (PRECISE)
O_c_2_e   (c,j,k,l,m, t,p)   operation cost per acre INCLUDING subsidy   (PRECISE)
O_c_s_2_e (c,j,k,l,m, t,p)   operation cost SUBSIDY per acre             (PRECISE)

Netrev_e  (c,j,k,l,m, t,p)   Net revenue per acre INCLUDING subsidy      (IMPRECISE)
Subacre_e (c,j,k,l,m, t,p)   SUBSIDY PER ACRE                            (IMPRECISE)
C_C_e     (c,j,k,l,m, t,p)   Capital cost per acre INCLUDING subsidy     (IMPRECISE)
*no C_c_s_e
O_C_e     (c,j,k,l,m, t,p)   Operations cost per acre INCLUDING subsidy  (IMPRECISE)
*no O_c_s_e

Incom_a_e(c,j,k,l,m, t,p)    Farm Income over acres for each index
Incom_c_e(  j,k,l,m, t,p)    Farm income over cons measures for each index
Income_e (           t,p)    Farm Income by period - with and without project

NPV_e    (             p)    NPV farm income WITHOUT and WITH project
NPV_wop_e                    NPV farm income WITHOUT project
NPV_wip_e                    NPV farm income WITH project

subsd_a_e(c,j,k,l,m, t,p)    Total taxpayer cost of subsidy over acres for each index
P_cost_e  (            p)    NPV taxpayer cost conservation program WITH and WITHOUT project
P_cost_wip_e                 NPV taxpayer cost conservation program WITH project
;

S_Water_e(    k,    t,p)..  sum((c,j,  l    )$jp(j,p),   Acres_v(c,j,k,l,'sw', t,p) * Wateruse_p(c,j,k,l,'sw' ,p))   =L= S_Waterrhs_p(    k,    t,  p);
Land_e   (    k,l,  t,p)..  sum((c,j,    m  )$jp(j,p),   Acres_v(c,j,k,l,   m, t,p) * Landuse_p (c,j,k,l,m,  t,p))   =L= Landrhs_p   (    k,l,  t,  p);
up_front(cc,j,k,l,m,tt,p)$jp(j,p)..                      Acres_v(cc,j,k,l,   m,tt,p)                                =L= Acres_v    (cc,j,k,l,m,'2010',p);
Season_e (c,j,k,l,m,t, p)$jp(j,p)..                      Acres_v (c,j,k,'su_fa',m,t,p) * (1/rat_pct_p(k))           =L= Acres_v (c,j,k,'sp_su',m,t,p)   ;

Netrv_2_e(c,j,k,l,m,t,p)$jp(j,p)..  (Netrev_v (c,j,k,l,m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= Netrv_2_v   (c,j,k,l,m, t, p);
Subac_2_e(c,j,k,l,m,t,p)$jp(j,p)..  (subacre_v(c,j,k,l,m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= Subac_2_v   (c,j,k,l,m, t, p);
C_c_2_e  (c,j,k,l,m,t,p)$jp(j,p).. (C_c_v  (c,j,k,l,   m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= C_c_2_v     (c,j,k,l,m, t, p);
C_c_s_2_e(c,j,k,l,m,t,p)$jp(j,p).. (C_c_s_v(c,j,k,l,   m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= C_c_s_2_v   (c,j,k,l,m, t, p);
O_c_2_e  (c,j,k,l,m,t,p)$jp(j,p).. (O_c_v  (c,j,k,l,   m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= O_c_2_v     (c,j,k,l,m, t, p);
O_c_s_2_e(c,j,k,l,m,t,p)$jp(j,p).. (O_c_s_v(c,j,k,l,   m,t,p) * Acres_v(c,j,k,l,m,t,p)) / ((Acres_v(c,j,k,l,m,t,p)+epsi)) =E= O_c_s_2_v   (c,j,k,l,m, t, p);

Netrev_e (c,j,k,l,m, t,p)$jp(j,p).. Netrev_p(c,j,k,l,   m,t,p) + Subacre_v(c,j,k,l,m,t,p)                               =E= Netrev_v    (c,j,k,l,m,t,  p);
Subacre_e(c,j,k,l,m, t,p)$jp(j,p)..                      C_c_s_v(c,j,k,l,m,t,p)   +  O_c_s_v(c,j,k,l,m,t,p)          =E= Subacre_v   (c,j,k,l,m,t,  p);
*no C_c_s_e
C_C_e    (c,j,k,l,m, t,p)$jp(j,p)..                        C_C_p(c,j,k,l,m,t,p)   -  C_c_s_v(c,j,k,l,m,t,p)          =E= C_C_v       (c,j,k,l,m,t,  p);
*no O_c_s_e
O_C_e    (c,j,k,l,m, t,p)$jp(j,p)..                        O_C_p(c,j,k,l,m,t,p)   -  O_c_s_v(c,j,k,l,m,t,p)          =E= O_C_v       (c,j,k,l,m,t,  p);

Incom_a_e(c,j,k,l,m, t,p)$jp(j,p)..                              Acres_v(c,j,k,l,m,t,p) *   netrev_v(c,j,k,l,m,t,p)          =E= Incom_a_v   (c,j,k,l,m, t, p);
Incom_c_e(  j,k,l,m, t,p)..  sum( c             $jp(j,p), Acres_v(c,j,k,l,m,t,p) *   netrev_v(c,j,k,l,m,t,p))        =E= Incom_c_v   (  j,k,l,m, t, p);
Income_e (           t,p)..  sum((c,j,k,l,m  )  $jp(j,p), Acres_v(c,j,k,l,m,t,p) *   netrev_v(c,j,k,l,m,t,p))        =E= Income_v    (           t, p);

NPV_e    (             p)..  sum(           t           , Income_v(         t,p)                          * DF(t))   =E= NPV_v       (              p);
```

```
NPV_wop_e..                                                                  NPV_v      (          'wop')            =E= NPV_wop_v                   ;
NPV_wip_e..                                                                  NPV_v      (          'wip')            =E= NPV_wip_v                   ;

Subsd_a_e(c,j,k,l,m, t,p)$jp(j,p)..                   Acres_v(c,j,k,l,m,t,p)  * subac_2_v(c,j,k,l,m,t,p)          =E= Subsd_a_v   (c,j,k,l,m, t, p);
P_cost_e (             p)..  sum((c,j,k,l,m,t)  $jp(j,p), Subsd_a_v(c,j,k,l, m, t, p)                     * DF(t))   =E= P_cost_v    (               p);
P_cost_wip_e..               sum((c,j,k,l,m,t)          , Subsd_a_v(c,j,k,l, m, t, 'wip')                 * DF(t))   =E= P_cos_wip_v                 ;

Model LCRA_TX /all/;

options nlp = conopt;


********

*--------------------------------------------------------------------------------------------------------------------
*      #1 baseline model run:  no farm subsidy no water losses from irrigated ag and no project                    *
*--------------------------------------------------------------------------------------------------------------------

* zero program cost and zero subsidy per acre in the base run

Subacre_v.fx   (c,j,k,l,m,t,p) = 0;

* simple net income maximization for base run

Solve LCRA_TX using nlp maximizing NPV_wop_v;

*--------------------------------------------------------------------------------------------------------------------
*      #2 model run:  reduce water supply but allow producers access to high yield rice                            *
*      check to see if income is as high or higher because of higher yielding rice varieties                       *
*      maximize income under with-project conditions                                                               *
*--------------------------------------------------------------------------------------------------------------------

* fix without project acres at above optimized levels
Acres_v.fx(c,j,k,l,m,t,'wop')   =  Acres_v.l(c,j,k,l,m,t,'wop');

options nlp = minos;

Solve LCRA_TX using nlp maximizing NPV_wip_v;


*--------------------------------------------------------------------------------------------------------------------
*      #3 model run:  if #2 produces income too low, require income to be at least as high as base                 *
*      by looking for a cost-minimizing subsidy                                                                     *
*--------------------------------------------------------------------------------------------------------------------

*Incom_c_v.lo( j,k,l,m,t,'wip') = (Incom_c_v.l( j,k,l,m,t,'wip'));

*Solve LCRA_TX using nlp minimizing P_cos_wip_v;

*$ontext


OPTION acres_v:0:0:1;        DISPLAY acres_v.L;
option income_v:0:0:1;       display income_v.l;
option p_cost_v:0:0:1;       display P_cost_v.l;

**************** Section 6 **************************************************************
*  Post-optimality output                                                                *
******************************************************************************************

******************************************************************************************
*  The following section displays input data used                                        *
******************************************************************************************
```

```
file  a_input_dat / a_input_data.txt/;
      put a_input_dat;
      put    '                          DRAFT:  JUNE 18 2006                                                    '/;
      put    'Input Data used for TX Gulf Farm Income Maximization Model: 2005 - 2010 With and Without Project'/;
      put /  'C_meas ', '   Crop   ',  '   District  ', '     Season  ', '  Source   ','Year  ', ' Policy',
             ' Rice_price '
             'WU_per_ac ' , ' Yld(cwt) ', ' TR_pr_ac', ' TC_pr_ac',  '  NR_pr_ac  ', ' Sup_Afeet',  ' Sup_Acres ' /;
      loop((c,j,k,l,m,t,p),
        put @1, c.tl, @10, j.tl, @22, k.tl, @37, l.tl, @47, m.tl, @55, t.tl, @62, p.tl,
            @68, Price_p   (  j,  l, t  ): 8:2,
                 Wateruse_p(c,j,k,l,m, p): 9:2,
                 Yield_p  (c,j,k,l,    p):10:0,
                 TotRev_p (c,j,k,l,  t,p):10:0,
                 TotCos_p (c,j,k,l,m,t,p):11:0,
                 netrev_p (c,j,k,l,m,t,p):10:0,
             S_Waterrhs_p (    k,    t,p):12:0,
                Landrhs_p (    k,l,  t,p):12:0
             /);

*******************************************************************************************
* Next are summary results of optimization model run                                      *
*******************************************************************************************

file  a_results / a_model_results.txt/;
      put a_results;
      put    '                          DRAFT:  JUNE 18 2006                                                '/;
      put    'Results of TX Gulf Farm Income maximization Model: 2005 - 2010 With and Without Project'     /;
      put /  'c_meas ', '   crop   ',  '   District  ', '     season  ', '  source   ','year  ', ' policy',
           ' TR_pr_ac', ' TC_pr_ac',  '  NR_pa_p  ', '  sub_ac_v', '  NR_pc_v','   acres','  af_pa','  ac_feet  ',
           ' t_income','  PV_income '/;

      loop((c,j,k,l,m,t,p),
        put @1, c.tl, @10, j.tl, @22, k.tl, @37, l.tl, @47, m.tl, @55, t.tl, @62, p.tl,
            @68, TotRev_p (c,j,k,l,  t,p): 8:0,
                 TotCos_p (c,j,k,l,m,t,p): 8:0,
                 netrev_p (c,j,k,l,m,t,p): 9:0,
               subac_2_v.l(c,j,k,l,m,t,p):10:0,
               Netrv_2_v.l(c,j,k,l,m,t,p): 9:0,
                 acres_v.l(c,j,k,l,m,t,p):12:0,
                Wateruse_p(c,j,k,l,m,  p): 6:2,
                (Acres_v.l(c,j,k,l,m,t,p) * Wateruse_p(c,j,k,l,m,p)): 9:0,
               Incom_a_v.l(c,j,k,l,m,t,p):11:0,
                   NPV_v.l(            p):11:0
            /);
*-----------------------------------------------------------------------------------------------------------------------
*                                             THE END                                                                  *
*-----------------------------------------------------------------------------------------------------------------------


*$offtext
```